
(Edisi V2.0-Update)

PROGRAM STUDI SISTEM KOMPUTER

FAKULTAS ILMU KOMPUTER

UNIVERSITAS NAROTAMA

2012
DISPLAY LED

TUJUAN:

1. Mahasiswa memahami rangkaian mikrokontroller untuk menghidupkan dan mematikan LED.

2. Mahasiswa dapat memahami program assembly untuk menghidupkan dan mematikan LED.

3. Mahasiswa memahami beberapa instruksi assembly dasar, MOV, Setb, Clr, RL dan RR, CJNE.

4. Mahasiswa memahami pembuatan instruksi waktu tunda
DASAR TEORI

Gambar 1.1. Diagram skematik IC89s51
Pada diagram skematik tersebut, terdapat 4 (empat) buah port, Port0, Port 1, Port 2 dan Port 3, yang masing-masing terdiri dari 8 bit data. Tata cara penulisan port adalah: P0.0 s/d P0.7, P1.0 s/d P1.7, P2.0 s/d P2.7 dan P3.0 s/d P3.7.
Instruksi MOV, adalah instruksi copy data 8 bit.

Contoh:

Mov P0,A ; copy isi data yang ada di akumulator ke Port 0

Mov P0,#01010101b; isi Port 0 dengan data biner 01010101b atau keluarkan data biner tersebut ke P0
Instruksi Setb, adalah instruksi mengirim logika 1 pada sebuah bit port

Instruksi Clr, adalah instruksi mengirim logika 0 pada sebut bit port

Contoh:

Setb P0.0; kirim logika 1 pada P0.0

Clr P0.7; kirim logika 0 pada P0.7

Instruksi RL, Geser data A ke kiri

Instruksi RR, Geser data A ke kanan

Contoh:

Mov A,#00000001b; isi data akumulator dengan data biner tersebut

RL A; Geser data yang ada di akumulator ke kiri, shg data akan berisi A=00000010b

RL A; Geser data yang ada di akumulator ke kiri, shg data akan berisi A=00000100b

RR A; Geser data yang ada di akumulator ke kanan, shg data akan berisi A=00000010b

Instruksi CJNE, Compare and Jump if Not Equal (bandingkan dan lompatlah jika tidak sama)

Contoh:

 Mov A, P2

Periksa: CJNE A,#1,Periksa

 Mov P0,A

Pada instruksi tersebut, copy data dari P2 ke A, bandingkan data A dengan 1 apabila tidak sama (A=/ 1 maka akan lompat ke label Periksa ). Apabila sama ( A=#1, maka akan turun ke bawah).
Instuksi DEC, Decrement register (mengurangi isi data register)

Instruksi INC, Increment register (menambah isi data register)

Contoh:

Mov R0,#10; isi R0 dengan data 10 (sepuluh)

DEC R0; kurangi isi R0 dengan 1, sehingga isi R0 sekarang menjadi 9(sembilan)

DEC R0; kurangi isi R0 dengan 1, sehingga isi R0 sekarang menjadi 8(delapan)

INC R0; tambahkan isi R0 dengan 1, sehingga isi R0 sekarang menjadi 9(sembilan)

Instruksi SJMP, short jump, melompat ke sebuah label dengan jarak dekat

Contoh:

Mulai1: mov P0,#01010101b

 Sjmp Mulai1

Pada perintah tersebut berarti, program akan melompat ke label Mulai1, ingat label tidak

diperkenankan ada spasi: Mulai 1
Instruksi CALL, memanggil sebuah subrutin

Contoh:

Ayo: 
mov P0,#01010101b

Call WaktuTunda

Sjmp Ayo

WaktuTunda: 
Bla

Bla

Bla

Ret

end

Pada perintah tersebut, program akan memanggil subrutin WaktuTunda, yang berisi instruksi

Bla-Bla-Bal, dan diakhiri dengan RET. Seperti halnya sebuah label, penamaan sebuah subrutin,

tidak diperkenankan ada spasi: Waktu Tunda
Instruksi DJNZ, Decrement and Jump if Not Zero

Mov R0,#10

AyoRek: DJNZ R0,AyoRek

End

Pada program tersebut data register R0, diisi dengan data 10, pada instruksi DJNZ, maka program akan mengurangi data pada register R0 (R0=R0-1), apabila data register R0 tidak nol (R0=/ 0) maka program akan melompat ke AyoRek.
Struktur dalam penulisan bahasa Asembly
Contoh:

AyoMulai: 
mov P0,#01010101b

Call Pekerjaan1

Call Pekerjaan2

Sjmp AyoMulai

;

Pekerjaan1: 
Bla

Bla

Ret

;

Pekerjaan2: 
Bla

Bla

ret

End
Pada contoh penulisan program assembly tersebut, tampak sebuah label AyoMulai (perhatikan tatacara pembuatan lebel ini, tanpa ada spasi dan boleh dalam bahasa indonesia, untuk keperluan lompatan, dan diakhiri dengan End. Perhatikan tataletak pembuatan subrutin tersebut.
Struktur progam yang salah

AyoMulai: 
mov P0,#01010101b

Sjmp AyoMulai

Call Pekerjaan1

Call Pekerjaan2

;

Pekerjaan1: 
Bla

Bla

Ret

;

Pekerjaan2: 
Bla

Bla

ret

End
Apabila instruksi tersebut dilakukan pada saat instruksi SJMP AyoMulai, maka pemanggilan

instruksi yang ada dibawahnya akan terabaikan atau tidak diproses.

Gambar 1.2 Rangkaian Display LED
Perhatikan pada gambar 1.1 tersebut. Delapan buah LED terhubung ke port 0, yang difungsikan sebagai output. Pada konfigurasi tersebut LED akan nyala bila diberi logika LOW ‘0’ melalui port 0, dan LED akan padam bila diberi logika HIGH ‘1’ melalui port 0. Agar Display LED ini dapat aktiv maka jumper EN LED harus dihubungkan singkat.
Percobaan 1.1. Menghidupkan LED dengan Instruksi MOV
Pada percobaan 1.1 ini LED akan dihidupkan atau dimatikan dengan mengirimkan data tertentu pada port 0. Untuk melakukan percobaan ini lakukan beberapa langkah sebagai berikut:

1. Hubungkan jumper pada LED_EN, untuk mengaktifkan 8 buah LED

2. Hubungkan modul Microcontroller Trainer dengan power supply +5V

3. Hubungkan modul Microcontroller Trainer dengan rangkaian programmer

4. Buka Program M-IDE Studio for MCS-51, sebagai editor dan compiler program
[image: image3.emf]
Gambar 1.3. Tampilan kosong MIDE Studio for MCS51

Gambar 1.4. Diagram alir percobaan 1.1

5. Ketik program berikut ini:
Org 0h

Start: Mov P0,#11110000b ; Isi P0 dengan 11110000

sjmp start ; lompat ke start

End
6. Simpanlah program yang anda ketik dan beri nama : prog11a.asm

7. Pada program MIDE tersebut pilih Build /F9 atau [image: image5.png]


untuk melakukan kompilasi program dari *.asm ke *.hex.

8. Lakukan pemrograman mikrokontroller dengan menggunakan Program ISP Software (Lihat Petunjuk Penggunaan)
[image: image6.emf]
Gambar 1.5. Tampilan program ISP Software
9. Lakukan pengamatan pada LED

10. Gantilah data #11110000b tersebut diatas dengan data seperti pada tabel berikut dan lakukan pencatatan LED mana yang padam.

D0 D1 D2 D3 D4 D5 D6 D7 LED YANG PADAM

Percobaan 1.2. Pembuatan Subrutine Waktu Tunda

Pada percobaan ini, 8 LED akan berkedip secara kontinu, yang diakibatkan oleh pemberian waktu tunda.

Untuk melakukan percobaan ini lakukan beberapa langkah sebagai berikut:
1. Hubungkan modul Microcontroller Trainer dengan power supply +5V

2. Hubungkan modul Microcontroller Trainer dengan rangkaian programmer

3. Buka Program M-IDE Studio for MCS-51, sebagai editor dan compiler program

Gambar 1.6. Diagram alir percobaan 1.2

4. Ketik program berikut ini:

             Org 0h

Start: 
Mov P0,#11111111b ; Kirim data biner 11111111 ke P0

Call Delay ; Memanggil waktu tunda

Mov P0,#00000000b ; Kirim data biner 00000000 ke P0

Call Delay ; Memanggil waktu tunda

Sjmp start ; lompat ke start

;

Delay: 
mov R1,#255 ; Isi 255 ke R1

Del1: 
mov R2,#255 ; Isi 255 ke R2

Del2: 
djnz R2,del2 ; Dec. R2 dan lompat ke del2 if NZ

djnz R1,del1

ret

end
5. Simpanlah program yang anda ketik dan beri nama : prog12a.asm

6. Pada program MIDE tersebut pilih Build /F9 atau [image: image9.png]


untuk melakukan kompilasi program dari *.asm ke *.hex.

7. Lakukan pemrograman mikrokontroller dengan menggunakan Program ISP Software (Lihat Petunjuk Penggunaan)

8. Lakukan pengamatan pada LED

9. Gantilah data tersebut untuk mengedipkan sebuah LED, dua buah LED dan seterusnya,sesuai tabel berikut ini.

D0 D1 D2 D3 D4 D5 D6 D7 LED YANG PADAM

Percobaan 1.3. Instruksi Setb dan Clr

Percobaan ini, sebuah LED atau lebih dapat dihidupkan atau dimatikan dengan perintah setb dan clr.

Untuk melakukan percobaan ini lakukan beberapa langkah sebagai berikut:

1. Hubungkan modul Microcontroller Trainer dengan power supply +5V

2. Hubungkan modul Microcontroller Trainer dengan rangkaian programmer

3. Buka Program M-IDE Studio for MCS-51, sebagai editor dan compiler program

4. Ketik program berikut ini:

Org 0h

Start: 
Clr P0.0 ; Kirim logika ‘0’ ke P0.0

Clr P0.1 ; Kirim logika ‘0’ ke P0.1

Clr P0.2 ; Kirim logika ‘0’ ke P0.2

Clr P0.3 ; Kirim logika ‘0’ ke P0.3

Clr P0.4 ; Kirim logika ‘0’ ke P0.4

Clr P0.5 ; Kirim logika ‘0’ ke P0.5

Clr P0.6 ; Kirim logika ‘0’ ke P0.6

Clr P0.7 ; Kirim logika ‘0’ ke P0.7

Call Delay ; Memanggil waktu tunda

Setb P0.0 ; Kirim logika ‘1’ ke P0.0

Setb P0.1 ; Kirim logika ‘1’ ke P0.1

Setb P0.2 ; Kirim logika ‘1’ ke P0.2

Setb P0.3 ; Kirim logika ‘1’ ke P0.3

Setb P0.4 ; Kirim logika ‘1’ ke P0.4

Setb P0.5 ; Kirim logika ‘1’ ke P0.5

Setb P0.6 ; Kirim logika ‘1’ ke P0.6

Setb P0.7 ; Kirim logika ‘1’ ke P0.7

Call Delay ; Memanggil waktu tunda

Sjmp start ; lompat ke start

;

Delay: 
mov R1,#255

Del1: 
mov R2,#255

Del2:
djnz R2,del2

djnz R1,del1

ret

end
5. Simpanlah program yang anda ketik dan beri nama : prog13a.asm

6. Pada program MIDE tersebut pilih Build /F9 atau [image: image11.png]


untuk melakukan kompilasi program dari *.asm ke *.hex.

7. Lakukan pemrograman mikrokontroller dengan menggunakan Program ISP Software (Lihat Petunjuk Penggunaan)

8. Lakukan pengamatan pada LED

9. Rencanakan program untuk mengedipkan dua buah LED, tiga buah LED dan seterusnya dengan instruksi Setb dan Clr.
Percobaan 1.4. LED Berjalan dengan Instruksi RL/ RR ( Rotate Left dan Right )

Pada percobaan ini, sebuah LED akan berjalan dari kiri ke kanan dan sebaliknya, program ini memanfaatkan instruksi RL dan RR.

Untuk melakukan percobaan ini lakukan beberapa langkah sebagai berikut:

1. Hubungkan modul Microcontroller Trainer dengan power supply +5V

2. Hubungkan modul Microcontroller Trainer dengan rangkaian programmer

3. Buka Program M-IDE Studio for MCS-51, sebagai editor dan compiler program

Gambar 1.6. Diagram alir percobaan 1.3
4. Ketik program berikut ini:
Org 0H

Start: 
Mov A,#11111110b;isi A dengan 11111110b

Mov R0,#7 ;isi R0 dengan 7

Kiri: 
Mov P0,A ;copy A ke P0

Call Delay ;panggil sub rutine delay

RL A ;data A diputar ke kiri

DEC R0 ;data R0 dikurangi 1

CJNE R0,#0,Kiri ;bandingkan R0 dg 0, if NE jump kiri

;

Mov R0,#7

Kanan: Mov P0,A

Call Delay

RR A ;
data A diputar ke kanan

DEC R0 ;data RO dikurangi 1

CJNE R0,#0,Kanan ;bandingkan R0 dg 0, if NE jump kanan

Sjmp Start

;

Delay: 
mov R1,#255

Del1: 
mov R2,#255

Del2: 
djnz R2,del2

djnz R1,del1

ret

end

5. Simpanlah program yang anda ketik dan beri nama : prog14a.asm

6. Pada program MIDE tersebut pilih Build /F9 atau [image: image13.png]


untuk melakukan kompilasi program dari *.asm ke *.hex.

7. Lakukan pemrograman mikrokontroller dengan menggunakan Program Downloader

8. Lakukan pengamatan pada LED
SAKLAR PUSH BUTTON
TUJUAN:

1 Mahasiswa memahami rangkaian mikrokontroller dengan interface ke saklar

2. Mahasiswa dapat memahami program assembly untuk mengambil data saklar dan mengeluarkan data ke LED.

3. Mahasiswa memahami beberapa instruksi assembly dasar, MOV, Setb, Clr, RL dan RR.

DASAR TEORI

Instruksi Mov A, P2

Contoh:

Mov A, P2; pada instruksi ini, data pada port P2 akan di copy ke register Akumulator

Instruksi JB, Jump if Bit set

Contoh:

MasihDiSini: JB P2.0, MasihDiSini

Call AyoBelajarRek

...............

...............

Pada instruksi tersebut, menyatakan bahwa bit P2.0 akan diuji, apakah berlogika 1(satu), bila ya maka program akan melompat ke label MasihDiSini, apabila bit P2.0 berlogika 0(nol), maka program akan melanjutkan instruksi dibawahnya (subrutin AyoBelajarRek)

GAMBAR:

( b )

Gambar 2.2. Rangkaian Interface (a) Push Button dan LED (b)
Pada gambar 2.2.a tersebut tampak rangkaian push button, bila saklar ditekan maka port sesuai dengan bit tersebut akan mendapat logika low ‘0’ dan sebaliknya bila saklar tidak ditekan maka port tersebut akan mendapat logika high ‘1’.
Percobaan 2.1. Mengambil Data Saklar dan Mengeluarkan ke LED
Pada percobaan ini, LED akan nyala bila saklar ditekan sesuai dengan bit saklar yang ditekan.

Untuk melakukan percobaan ini lakukan beberapa langkah sebagai berikut:

1. Buka jumper pada EN_DAC apabila sedang terhubung.

2. Hubungkan modul Microcontroller Trainer dengan power supply +5V

3. Hubungkan modul Microcontroller Trainer dengan rangkaian programmer

4. Buka Program M-IDE Studio for MCS-51, sebagai editor dan compiler program

Gambar 2.3. Diagram alir percobaan 2.1

5. Ketik program berikut ini:

Org 0h

Start: 
Mov A, P2 ;copy data P2 dan Simpan ke A

Mov P0, A ;Kirim data A ke P0

sjmp start ;lompat ke start

end

6. Simpanlah program yang anda ketik dan beri nama : prog21a.asm

7. Pada program MIDE tersebut pilih Build /F9 atau [image: image17.png]


untuk melakukan kompilasi program dari *.asm ke *.hex.

8. Lakukan pemrograman mikrokontroller dengan menggunakan Program ISP Software (Lihat Petunjuk Penggunaan)

9. Lakukan pengamatan pada LED.

Percobaan 2.2. Mengatur Arah Gerak LED Ke Kanan-Kiri dengan Saklar Push Button P2.0 dan P2.1

Pada percobaan ini, LED akan berjalan kekanan atau kekiri mengikuti penekanan tombol P2.0 (arah kiri) atau P2.1 (arah kanan).

Untuk melakukan percobaan ini lakukan beberapa langkah sebagai berikut:

1. Hubungkan modul Microcontroller Trainer dengan power supply +5V

2. Hubungkan modul Microcontroller Trainer dengan rangkaian programmer

3. Buka Program M-IDE Studio for MCS-51, sebagai editor dan compiler program

Gambar 2.4. Diagram alir percobaan 2.2

4. Ketik program berikut ini:

org 0h

CekP20: 
JB P2.0,CekP21 ; Menguji bit P2.0, apakah adalah logika '1'

call RLeft ; Jika P2.0=0, maka LED putar kiri

; sjmp CekP20 ; jump forever to CekP20

CekP21: 
JB P2.1,CekP20 ; Menguji bit P2.1, apakah adalah logika '1'

call RRight ; Jika P2.1=0, maka LED putar kanan

sjmp CekP20 ; jump forever to CekP2.0

;======================================================

;Subrutin ini digunakan untuk menggerakkan LED Ke Kanan

;======================================================

RLeft: 

mov A,#11111110b ;send data 11111110 to Acc

RLeft1: 
mov P0,A ;send data A to P0

call delay ;call delay time

JB P2.1,RLeft2 ;Menguji bit P2.1, apakah adalah logika '1'

sjmp EndRLeft ;Jika P2.1=0, maka lompat ke EndRLeft

RLeft2: 
RL A ;data A di putar ke kiri

sjmp RLeft1 ;lompat ke Rleft1

EndRLeft:
 ret

;

;====================================================

;Subrutin ini digunakan untuk menggerakkan LED Ke Kiri

;====================================================

RRight: 
mov A,#01111111b ;send data to Acc

RRight1: 
mov P0,A ;send data to P0

call delay ;call delay time

JB P2.0,RRight2 ;Menguji bit P2.0, apakah adalah logika '1'

sjmp EndRRight ;Jika P2.0=0, maka lompat ke EndRRight

RRight2: 
RR A ;data A putar ke kanan

sjmp RRight1 ;lompat ke Rright1

EndRRight: 
ret

;=============================================

;subroutine delay time

;=============================================

delay: 
mov R1,#255

del1: 
mov R2,#255

del2: 
djnz R2,del2

djnz R1,del1

ret

end
5. Simpanlah program yang anda ketik dan beri nama : prog22a.asm

6. Pada program MIDE tersebut pilih Build /F9 atau [image: image20.png]


untuk melakukan kompilasi program dari *.asm ke *.hex.

7. Lakukan pemrograman mikrokontroller dengan menggunakan Downloader Software (Lihat Petunjuk Penggunaan)
Soal Tantangan:

1. Lakukan modifikasi pada percobaan 2.2. tersebut sesuai dengan dengan mengganti port dengan P2.2 (putar kanan ) dan P2.3 ( putar kiri ).

2. Buatlah suatu percobaan untuk menghidupkan dan mematikan sebuah led yang terhubung ke P0.0, dengan menggunakan sebuah saklar P2.0, sehingga apabila saklar ditekan sekali, LED nyala, dan ditekan berikutnya maka LED akan padam.

Percobaan 2.3. Setting Up/Dn dan Enter dengan display LED

Pada percobaan ini, LED yang ON akan bertambah atau berkurang mengikuti penekanan tombol P2.1 (UP) atau P2.2 (DN). LED akan berhenti bertambah atau atau berkurang bila ditakan tombol P2.0 (ENTER).

Untuk melakukan percobaan ini lakukan beberapa langkah sebagai berikut:

1. Hubungkan modul Microcontroller Trainer dengan power supply +5V

2. Hubungkan modul Microcontroller Trainer dengan rangkaian programmer

3. Buka Program M-IDE Studio for MCS-51, sebagai editor dan compiler program

Gambar 2.5. Diagram alir percobaan 2.3

4. Ketik program berikut ini:

org 0h

start:

mov R7,#1 ;inisialisasi data setting

Setup: 
mov A,R7 ;simpan data R7 ke A

cpl A ;komplemen A,

mov P0,A ;output data ke LED

jnb p2.0,getout;bila sw1(P2.0) ditekan mk getout(selesai)

jb P2.1,SetDn ;bila sw2(P2.1) ditekan mk INC R7

inc R7 ;R7:=R7+1

acall delay ;waktu tunda, antar penekanan tombol

cjne R7,#100d,setup;deteksi apakah setting=100d

mov R7,#1 ;reset R7 -> 1

sjmp Setup

;

SetDn: Mov A,R7 ;simpan data R7 ke A

cpl A ;komplemen A,

mov P0,A ;output data ke LED

jnb P2.0,getout;bila sw1(P2.0) ditekan mk getout(selesai)

jb p2.2,Setup ;bila sw2(P2.1) ditekan mk INC R7

dec R7 ;R7:=R7-1

acall delay ;waktu tunda lama penekanan tombol

cjne R7,#0d,setDn;deteksi apakah setting=0d

mov R7,#1d ;reset R7 -> 1

sjmp Setdn

getout:

sjmp getout

;

delay: 
mov R0,#255

delay1: mov R2,#255

djnz R2,$

djnz R0,delay1

ret

end
5. Simpanlah program yang anda ketik dan beri nama : prog23a.asm

6. Pada program MIDE tersebut pilih Build /F9 atau [image: image22.png]


untuk melakukan kompilasi program dari *.asm ke *.hex.

7. Lakukan pemrograman mikrokontroller dengan menggunakan Program ISP Software ( Lihat Petunjuk Penggunaan)

8. Lakukan pengamatan pada LED.
LCD KARAKTER

TUJUAN:

1. Mahasiswa memahami rangkaian interface mikrokontroller dengan LCD Karakter 2 x 16

2. Mahasiswa dapat memahami program assembly untuk menampilkan data ke LCD Karakter 2 x 16

3. Mahasiswa memahami beberapa instruksi assembly dasar, MOV, Setb, Clr, dan waktu tunda.

4. Mahasiswa memahami mencetak karakter pada posisi baris dan kolom tertentu

DASAR TEORI

[image: image23.png]Potensio Multitum
10k

LCD Character

P0.7..P0.0




Gambar 4.1. Rangkaian interface ke LCD Karakter 2 x16

Modul LCD Character dapat dengan mudah dihubungkan dengan mikrokontroller seperti AT89S51. LCD yang akan kita praktikumkan ini mempunyai lebar display 2 baris 16 kolom atau biasa disebut sebagai LCD Character 2x16, dengan 16 pin konektor, yang didifinisikan sebagai berikut:

Gambar 4.2. Modul LCD Karakter 2x16

Tabel 4.1 Pin dan Fungsi

[image: image25.png]PIN [ Name Function
1 Vs |Ground voltage
2 Vee 45V
3 Ve [Contrast voltage
Register Select
4 RS = Instruction Register
I = Data Register
Read/ Write, to choose write or read mode
5 RW = write mode
1 = read mode
Enable
6 E = start to lacht data to LCD character
1= disable
7 DB0  [LSB
8 DBI
DB2
10 DB3
11 DB4
12 DBS
13 DB6
14 DB7  |MSB
15 BPL  |Back Plane Light
16 GND  [Ground voltage





Display karakter pada LCD diatur oleh pin EN, RS dan RW:

Jalur EN dinamakan Enable. Jalur ini digunakan untuk memberitahu LCD bahwa anda sedang mengirimkan sebuah data. Untuk mengirimkan data ke LCD, maka melalui program EN harus dibuat logika low “0” dan set pada dua jalur kontrol yang lain RS dan RW. Ketika dua jalur yang lain telah siap, set EN dengan logika “1” dan tunggu untuk sejumlah waktu tertentu (sesuai dengan datasheet dari LCD tersebut ) dan berikutnya set EN ke logika low “0” lagi.

Jalur RS adalah jalur Register Select. Ketika RS berlogika low “0”, data akan dianggap sebagi sebuah perintah atau instruksi khusus ( seperti clear screen, posisi kursor dll ). Ketika RS berlogika high “1”, data yang dikirim adalah data text yang akan ditampilkan pada display LCD. Sebagai contoh, untuk menampilkan huruf “T” pada layar LCD maka RS harus diset logika high “1”.

Jalur RW adalah jalur kontrol Read/ Write. Ketika RW berlogika low (0), maka informasi pada bus data akan dituliskan pada layar LCD. Ketika RW berlogika high ”1”, maka program akan melakukan pembacaan memori dari LCD. Sedangkan pada aplikasi umum pin RW selalu diberi logika low ”0”.

Pada akhirnya, bus data terdiri dari 4 atau 8 jalur ( bergantung pada mode operasi yang dipilih oleh user). Pada kasus bus data 8 bit, jalur diacukan sebagai DB0 s/d DB7. Beberapa perintah dasar yang harus dipahami adalah inisialisasi LCD Character,

Function Set

Mengatur interface lebar data, jumlah dari baris dan ukuran font karakter

CATATAN:

X : Don’t care

DL: Mengatur lebar data

DL=1, Lebar data interface 8 bit ( DB7 s/d DB0)

DL=0, Lebar data interface 4 bit ( DB7 s/d DB4)

Ketika menggunakan lebar data 4 bit, data harus dikirimkan dua kali

N=1, Display dua baris

N=0, Display satu baris

Entry Mode Set

Mengatur increment/ decrement dan mode geser

Catatan:

I/D: Increment/ decrement dari alamat DDRAM dengan 1 ketika kode karakter dituliskan ke

DDRAM.

I/D = “0”, decrement

I/D= “1”, increment

S: Geser keseluruhan display kekanan dan kekiri

S=1, geser kekiri atau kekanan bergantung pada I/D

S=0, display tidak bergeser

Display On/ Off Cursor

Mengatur status display ON atau OFF, cursor ON/ OFF dan fungsi Cursor Blink

D : Mengatur display

D = 1, Display is ON

D = 0, Display is OFF

Pada kasus ini data display masih tetap berada di DDRAM, dan dapat ditampilkan kembali secara langsung dengan mengatur D=1.

C : Menampilkan kursor

C = 1, kursor ditampilkan

C = 0, kursor tidak ditampilkan

B : Karakter ditunjukkan dengan kursor yang berkedip

B=1, kursor blink

Clear Display

Perintah ini hapus layar

Geser Kursor dan Display

Geser posisi kursor atau display ke kanan atau kekiri tanpa menulis atau baca data display.

Fungsi ini digunakan untuk koreksi atau pencarian display

Catatan : x = Dont care

Memori LCD

1. DDRAM ( Display Data RAM )

Memori DDRAM digunakan untuk menyimpan karakter yang akan ditampilkan. Semua teks yang kita tuliskan ke modul LCD adalah disimpan didalam memory ini, dan modul LCD secara berturutan membaca memory ini untuk menampilkan teks ke modul LCD itu sendiri.

Gambar 4.3. Lokasi memori display LCD Karakter

Pada peta memori tersebut, daerah yang berwarna kuning ( 00 s/d 0F dan 40 s/d 4F ) adalah display yang tampak. Sebagaimanan yang anda lihat, jumlahnya sebanyak 16 karakter per baris dengan dua baris. Angka pada setiap kotak adalah alamat memori yang bersesuaian dengan posisi dari layar. Demikianlah karakter pertama di sudut kiri atas adalah menempati alamah 00h. Posisi karakter berikutnya adalah alamat 01h dan seterusnya.

Akan tetapi, karakter pertama dari baris 2 sebagaimana yang ditunjukkan pada peta memori adalah pada alamat 40h. Dimikianlah kita perlu untuk mengirim sebuah perintah ke LCD untuk mangatur letak posisi kursor pada baris dan kolom tertentu. Instruksi Set Posisi Kursor adalah 80h. Untuk ini kita perlu menambahkan alamat lokasi dimana kita berharap untuk menempatkan kursor.Sebagai contoh, kita ingin menampilkan kata ”World” pada baris ke dua pada posisi kolom ke sepuluh. Sesuai peta memori, posisi karakter pada kolom 11 dari baris ke dua, mempunyai alamat 4Ah, sehingga sebelum kita tulis kata ”World” pada LCD, kita harus mengirim instruksi set posisi kursor, dan perintah untuk instruksi ini adalah 80h ditambah dengan alamat 80h+4Ah =0Cah. Sehingga dengan mengirim perintah Cah ke LCD, akan menempatkan kursor pada baris kedua dan kolom ke 11 dari DDRAM.

2. CGROM ( Character Generator ROM )

Sebuah peta karakter yang dapat ditampilkan, sesuai dengan lokasi dari masingmasing

karakter.

INISIALISASI

Sebelum kita dapat menggunakan modul LCD, kita harus melakukan inisialisasi dan mengkonfigurasikannya. Hal ini dijalankan dengan mengirimkan sejumlah instruksi ke LCD. Antara lain: pengaturan lebar data interface 8 bit atau 4 bit data bus, pemilihan ukuran font karakter 5x8 atau 5x7 dan lain-lain, dengan instruksi sebagai berikut.

Init_lcd:

mov r1,#00000001b ;Display clear

call write_inst ;EN=pulse dan RS=0

mov r1,#00111000b ;Function set, ;Data 8 bit,2 line font 5x7

call write_inst ;Set bit EN=pulse dan RS=0

mov r1,#00001100b ;Display on, ;cursor off,cursor blink off

call write_inst ;EN=pulse dan RS=0

mov r1,#00000110b ;Entry mode, Set increment

call write_inst

ret

Percobaan 4.1. Tulis Sebuah Karakter pada LCD Karakter

Pada percobaan ini, karakter ‘A’ akan ditampilkan pada LCD Karakter

Untuk melakukan percobaan ini lakukan beberapa langkah sebagai berikut:

1. Pasang jumper LCD_EN, yang berfungsi untuk memberikan power supply +5V

2. Hubungkan modul Microcontroller Trainer dengan power supply +5V

3. Hubungkan modul Microcontroller Trainer dengan rangkaian programmer

4. Buka Program M-IDE Studio for MCS-51, sebagai editor dan compiler program

[image: image34.png](_Stat )

—r

Inisialisasi LCD

Start

Display Clear
R1 € 400000001

¥
Pilih DDRAM Add.= 0
R1 ¢ 80h
¥ Function Set

Tulis Instruksi

Data 8 bit, 2 line, 5x7 font
R1 € #00111000b

/" Cetak Karakter A
R1 €A
¥ Display on
Tulis Data Cursor off, blink off
R1 € #00001100b
P T v
( End J Tulis Instruksi
¥
Entry Mode.

(a)

Set Increment DDRAM, No Geser
R1 € #00000110b

[]

Tulis Instruksi

R

() (_End





Gambar 4.4. Diagram alir (a) rutin utama cetak karakter “A”, (b) subrutin inisialisasi LCD karakter pada percobaan 4.1

5. Ketik program berikut ini:

org 0h

call init_LCD

start: 
mov R1,#80h ; Lokasi Display RAM, Row=1 Col=1

call write_inst ; panggil tulis instruksi

mov R1,#'A' ; Cetak Karakter A

call write_data ; panggil tulis data

stop: 
sjmp stop

Init_lcd:

mov r1,#00000001b ;Display clear

call write_inst

mov r1,#00111000b ;Function set,Data 8 bit,2 line font 5x7

call write_inst

mov r1,#00001100b ;Display on, ;cursor off,cursor blink off

call write_inst

mov r1,#00000110b ;Entry mode, Set increment

call write_inst

ret

;

Write_inst:

clr P3.6 ; P3.6 = RS =0

mov P0,R1 ; copy R1 ke P0

setb P3.7 ; P3.7 =EN = 1

call delay ; call delay time

clr P3.7 ; P3.7 =EN = 0

ret

;

Write_data:

setb P3.6 ; P3.6 = RS =1

mov P0,R1 ; copy R1 ke P0

setb P3.7 ; P3.7 =EN = 1

call delay ; call delay time

clr p3.7 ; P3.7 =EN = 0

ret

;

delay: 
mov R0,#0

delay1:mov R7,#0fh

djnz R7,$

djnz R0,delay1

ret

end

6. Simpanlah program yang anda ketik dan beri nama : prog41a.asm

7. Pada program MIDE tersebut pilih Build /F9 untuk melakukan kompilasi program dari *.asm ke *.hex.

8. Lakukan pemrograman mikrokontroller dengan menggunakan Program ISP Software ( Lihat Petunjuk Penggunaan)

9. Lakukan pengamatan pada karakter yang tercetak pada LCD.
10. Lakukan modifikasi pada program tersebut untuk mencetak karakter lain, sesuai tabel:

Percobaan 4.2. Tulis Delapan Karakter pada LCD Karakter

Pada percobaan ini, kalimat ” Welcome” akan ditampilkan pada LCD Karakter pada baris 1 dan colom 1

Untuk melakukan percobaan ini lakukan beberapa langkah sebagai berikut:

1. Pasang jumper LCD_EN, yang berfungsi untuk memberikan power supply +5V

2. Hubungkan modul Microcontroller Trainer dengan power supply +5V

3. Hubungkan modul Microcontroller Trainer dengan rangkaian programmer

4. Buka Program M-IDE Studio for MCS-51, sebagai editor dan compiler program

5. Ketik program berikut ini:

org 0h

call init_LCD

start: 
mov R1,#80h ; Lokasi Display RAM, Row=1 Col=1

call write_inst

mov R1,#'W'

call write_data

mov R1,#'e'

call write_data

mov R1,#'l'

call write_data

mov R1,#'c'

call write_data

mov R1,#'o'

call write_data

mov R1,#'m'

call write_data

mov R1,#'e'

call write_data

stop: 
sjmp stop

Init_lcd:

mov r1,#00000001b ;Display clear

call write_inst

mov r1,#00111000b ;Function set, Data 8 bit,2 line font 5x7

call write_inst

mov r1,#00001100b ;Display on, cursor off,cursor blink off

call write_inst

mov r1,#00000110b ;Entry mode, Set increment

call write_inst

ret

;
Write_inst:

clr P3.6 ; P3.6 = RS =0

mov P0,R1 ; copy R1 ke P0

setb P3.7 ; P3.7 =EN = 1

call delay ; call delay time

clr P3.7 ; P3.7 =EN = 0

ret

;

Write_data:

setb P3.6 ; P3.6 = RS =1

mov P0,R1 ; copy R1 ke P0

setb P3.7 ; P3.7 =EN = 1

call delay ; call delay time

clr p3.7 ; P3.7 =EN = 0

ret

;

delay: 
mov R0,#0

delay1:mov R7,#0fh

djnz R7,$

djnz R0,delay1

ret

end
6. Simpanlah program yang anda ketik dan beri nama : prog42a.asm

7. Pada program MIDE tersebut pilih Build /F9 untuk melakukan kompilasi program dari *.asm ke *.hex.

8. Lakukan pemrograman mikrokontroller dengan menggunakan Program ISP Software (Lihat Petunjuk Penggunaan)

9. Lakukan pengamatan pada display LCD Karakter

10.Lakukan modifikasi pada program tersebut untuk mencetak karakter lain, sesuai tabel:

Percobaan 4.3. Tulis Karakter Dengan Look Up Table

Pada percobaan ini, kalimat ”Welcome home” akan ditampilkan pada layar LCD Karakter dengan menggunakan look up table.

Untuk melakukan percobaan ini lakukan beberapa langkah sebagai berikut:

1. Pasang jumper LCD_EN, yang berfungsi untuk memberikan power supply +5V

2. Hubungkan modul Microcontroller Trainer dengan power supply +5V

3. Hubungkan modul Microcontroller Trainer dengan rangkaian programmer

4. Buka Program M-IDE Studio for MCS-51, sebagai editor dan compiler program
5. Ketik program berikut ini:

org 0h

call init_LCD

start: 
call write_char

stop: 
sjmp stop

;

write_char:

mov dptr,#word1 ;DPTR = [ address word1 ]

mov r3,#16 ;R3=16,number character to be display

mov r1,#80h ;R1=80h,address DDRAM start position

call write_inst

;

write1:clr a ; A = 0

movc a,@a+dptr ; A = [A+ DPTR]

mov r1,A ; R1 = A

inc dptr ; DPTR = DPTR +1

call write_data;

djnz r3,write1 ; R3 = R3-1,

ret

;

Init_lcd:

mov r1,#00000001b ;Display clear

call write_inst ;

mov r1,#00111000b ;Function set, Data 8 bit,2 line font 5x7

call write_inst ;

mov r1,#00001100b ;Display on, cursor off,cursor blink off

call write_inst

mov r1,#00000110b ;Entry mode, Set increment

call write_inst

ret

;

Write_inst:

clr P3.6 ; P3.6 = RS =0

mov P0,R1 ; copy R1 ke P0

setb P3.7 ; P3.7 =EN = 1

call delay ; call delay time

clr P3.7 ; P3.7 =EN = 0

ret

;

Write_data:

setb P3.6 ; P3.6 = RS =1

mov P0,R1 ; copy R1 ke P0

setb P3.7 ; P3.7 =EN = 1

call delay ; call delay time

clr p3.7 ; P3.7 =EN = 0

ret

;
delay: 
mov R0,#0

delay1:mov R7,#0fh

djnz R7,$

djnz R0,delay1

ret

;

word1: DB ' Welcome Home '; Karakter yang disimpan di ROM

end

6. Simpanlah program yang anda ketik dan beri nama : prog43a.asm

7. Pada program MIDE tersebut pilih Build /F9 untuk melakukan kompilasi program dari *.asm ke *.hex.

8. Lakukan pemrograman mikrokontroller dengan menggunakan Program ISP Software (Lihat Petunjuk Penggunaan)

9. Lakukan pengamatan pada LCD.

10. Lakukan modifikasi pada program tersebut untuk mencetak karakter lain, sesuai tabel:

Soal Tantangan:

1. Dengan memanfaatkan instruksi shift display dan cursor, cetaklah kata Surabaya yang dapat bergeser kekanan sebanyak 10 x dan bergeser kekiri sebanyak 10x dan berhenti.

2. Buatlah interaksi geser karakter ke kanan dan ke kiri, dengan menggunakan tombol P2.0, P2.1, untuk mengatur geser ke kanan dan ke kiri, untuk dengan kata Surabaya
Soal Kuis:

1. Apakah yang dimaksud dengan 2 x 16 pada LCD Karakter 2x16

2. Pelajari tentang pin konfigurasi, dan fungsi dari LCD Karakter

3. Pelajari tentang Function Set, Entry Mode, Display On/Off, Shift Cursor/Display

4. Pelajari rangkaian lengkap interfacing LCD ke Mikrokontroller

ANALOG TO DIGITAL CONVERTER ( ADC )

TUJUAN:

1. Mahasiswa memahami rangkaian interface mikrokontroller dengan ADC 0804

2. Mahasiswa memahami setting tegangan referensi Vref ADC0804

3. Mahasiswa memahami perhitungan tegangan resolusi ADC0804

4. Mahasiswa dapat memahami program assembly untuk menampilkan data ADC ke 7 Segmen

5. Mahasiswa dapat memahami prog. assembly untuk menampilkan data ADC ke LCD Karakter 2 x 16

DASAR TEORI

Gambar 5.1 Rangkaian ADC0804

Konverter A/D tersedia secara komersial sebagai rangkaian terpadu dengan resolusi 8 bit sampai dengan 16 bit. Pada percobaan ini akan memperkenalkan ADC0801, yaitu sebagai sebuah konverter A/D 8 bit yang mudah diinterfacekandengan sistem mikrokontroller. A/D ini menggunakan metode approksimasi berturut-turut untuk mengkonversikan masukan analog (0-5V) menjadi data digital 8 bit yang ekivalen. ADC0801 mempunyai pembangkit clock internal dan memerlukan catu daya +5V dan mempunyai waktu konversi optimum sekitar 100us.

[image: image39.emf]
Gambar 5.2 Konfigurasi pin ADC0804

Diagram konfigurasi pin ADC0804 ditunjukkan pada gambar 5.2. Pin 11 sampai 18 ( keluaran digital ) adalah keluaran tiga keadaan, yang dapat dihubungkan langsung dengan bus data bilamana diperlukan. Apabila CS ( pin 1 ) atau RD (pin2) dalam keadaan high (“1”), pin 11 sampai 18 akan mengambang (high impedanze ), apabila CS dan RD rendah keduanya, keluaran digital akan muncul pada saluran keluaran.

Sinyal mulai konversi pada WR (pin 3). Untuk memulai suatu konversi, CS harus rendah. Bilamana WR menjadi rendah, konverter akam mengalami reset, dan ketika WR kembali kepada keadaan high, konversi segera dimulai.

Konversi detak konverter harus terletak dalam daereh frekuensi 100 sampai 800kHz. CLK IN (pin 4) dapat diturunkan dari detak mikrokontroller, sebagai kemungkinan lain, kita dapat mempergunakan pembangkit clock internal dengan memasang rangkaian RC antara CLN IN (pin 4) dan CLK R ( pin 19).

Pin 5 adalah saluran yang digunakan untuk INTR, sinyal selesai konversi. INTR akan menjadi tinggi pada saat memulai konversi, dan akan aktiv rendah bila konversi telah selesai. Tepi turun sinyal INTR dapat dipergunakan untuk menginterupsi sistem mikrokontroller, supaya mikrokontroller melakukan pencabangan ke subrutine pelayanan yang memproses keluaran konverter.

Pin 6 dan 7 adalah masukan diferensial bagi sinyal analog. A/D ini mempunyai dua ground, AGND (pin 8) dan D GND ( pin10). Kedua pin ini harus dihubungkan dengan ground. Pin 20 harus dihubungkan dengan catu daya +5V A/D ini mempunyai dua buah ground, A GND ( pin 8 ) dan D GND ( pin 10). Keduanya harus dihubungkan dengan catu daya, sebesar +5V.

Pada A/D 0804 V REF  merupakan tegangan referensi yang digunakan untuk offset suatu keluaran digital maksimum. Dengan persamaan sebagai berikut:

Misalnya anda menginginkan masuk analog maksimum sebesar 4 V, maka:

Resolusi ini mempunyai arti sebagai berikut:

A/D ini dapat dirangkai untuk menghasilkan konversi secara kontinu. Untuk melaksanakannya, kita harus menghubungkan CS, dan RD ke ground dan menyambungkan WR dengan INTR seperti pada gambar dibawah ini. Maka dengan ini keluaran digital yang kontinu akan muncul, karena sinyal INTR menggerakkan masukan WR. Pada akhir konversi INTR berubah menjadi low, sehingga keadaan ini akan mereset konverter dan mulai konversi.

Tabel 5.1 Koneksi Interface ADC ke Mikrokontroller

Tabel 5.2. Instruksi logika pada pin kontrol A/D 0804

Percobaan 5.2. ADC0804 dan Display ke LCD Karakter 2x16

Pada percobaan ini, Data ADC dalam desimal akan ditampilkan pada LCD Karakter 2x16 pada Baris 1, Colom 1, 2 dan 3, yang masing-masing menampilkan data ratusan, puluhan dan satuan.

Gambar 5.8. Rangkaian Interface LCD Karakter

Untuk melakukan percobaan ini lakukan beberapa langkah sebagai berikut:

1. Pada saat langkah pemrograman posisikan saklar togle ke posisi PROG

2. Posisikan saklar togle ke RUN untuk mengaktifkan ADC0804 CS=0

3. Hubungkan modul Microcontroller Trainer dengan power supply +5V

4. Hubungkan modul Microcontroller Trainer dengan rangkaian programmer

5. Buka Program M-IDE Studio for MCS-51, sebagai editor dan compiler program

Gambar 5.9 Diagram alir utama percobaan 5.2

6. Ketik program berikut ini:

org 0h

ratusan equ 30h

puluhan equ 31h

satuan equ 32h

;

org 0h

call init_LCD

call write_char

start: 
call ADC

call Bin2Dec

call Write2LCD

sjmp start

;

;===============================================

;Subrutin ini digunakan untuk mengambil data ADC MUX X0

;===============================================

ADC: 
clr P2.6

clr P2.7

clr P3.3

nop

nop

nop

setb P3.3

eoc: 
jb P3.2,eoc

clr P3.4

mov A,P1

setb P3.4

ret

;

;=================================================

;Subrutin untuk menampilkan data ke LCD character 2 x16

;pada DDRAM 0C9 0CA 0CB untukratusan, puluhan, and satuan

;=================================================

Write2LCD:

mov r1,#0c9h

call write_inst

mov a,ratusan

add a,#30h

mov r1,a

call write_data

;

mov r1,#0cah

call write_inst

mov a,puluhan

add a,#30h

mov r1,a

call write_data

;

mov r1,#0cbh

call write_inst

mov a,satuan

add a,#30h

mov r1,a

call write_data

ret

;

;=======================================

;Subrutin ini untuk merubah data biner ke desimal

;menjadi 3 digit = ratusan-puluhan-satuan

;=======================================

Bin2Dec:

mov b,#100d

div ab

mov ratusan,a

mov a,b

mov b,#10d

div ab

mov puluhan,a

mov satuan,b

ret

;

;===============================================

;Subrutin untuk menampilkan tulisan Data ADC0804

; pada baris 1

;===============================================

write_char:

mov dptr,#word1 ;DPTR = [ address word1 ]

mov r3,#16 ;R3=16,number character to be display

mov r1,#80h ;R1=80h,address DDRAM start position

acall write_inst

;

write1:clr a ; A = 0

movc a,@a+dptr ; A = [A+ DPTR]

mov r1,A ; R1 = A

inc dptr ; DPTR = DPTR +1

acall write_data ;

djnz r3,write1 ; R3 = R3-1,

ret

;

Init_lcd:

mov r1,#00000001b ;Display clear

call write_inst

mov r1,#00111000b ;Function set,Data 8 bit,2 line font 5x7

call write_inst

mov r1,#00001100b ;Display on, cursor off,cursor blink off

call write_inst
mov r1,#00000110b ;Entry mode, Set increment

call write_inst

ret

;

write_inst:

clr P3.6 ; RS = P2.0 = 0, write mode instruction

mov P0,R1 ; D7 s/d D0 = P0 = R1

setb P3.7 ; EN = 1 = P2.1

call delay ; call delay time

clr P3.7 ; EN = 0 = P2.1

ret

;

Write_data:

setb P3.6 ; RS = P2.0 = 1, write mode data

mov P0,R1 ; D7 s/d D0 = P0 = R1

setb P3.7 ; EN = 1 = P2.1

call delay ; call delay time

clr p3.7 ; EN = 0 = P2.1

ret

;

delay: 
mov R0,#0

delay1:mov R2,#0fh

djnz R2,$

djnz R0,delay1

ret

;

word1: DB ' Data ADC0804 '

end

7. Simpanlah program yang anda ketik dan beri nama : prog52a.asm

8. Pada program MIDE tersebut pilih Build /F9 atau untuk melakukan kompilasi program dari *.asm ke *.hex.

9. Lakukan pemrograman mikrokontroller dengan menggunakan Program ISP Software (Lihat Petunjuk Penggunaan)

10. Lakukan modifikasi pada program tersebut dengan manambahkan kata SUHU: , pada Baris 2 diikuti dengan data ADC.

Percobaan 5.3. Aplikasi program komparator dengan memanfaatkan instruksi aritmatika dan instruksi lompatan untuk pengaturan suhu dengan display LCD Karakter

Dalam dunia elektronika, rangkaian komparator, umumnya diwujudkan dengan memanfaatkan rangkaian op-amp yang dibangun sebagai komparator. Sesuai dengan prinsip kerja komparator, membandingkan dua buah tegangan yang masuk pada input INV dan NON INV, untuk menghasilkan suatu output tegangan saturasi. Dengan memanfaatkan instruksi aritmatika SUBB dan instruksi lompatan JZ dan JC, maka rangkaian analog ini dapat digantikan dengan menggunakan pemrograman assembly.

Gambar 5.10 Rangkaian komparator analog dengan IC OP-AMP
Apabila tegangan yang masuk pada VREF lebih besar daripada tegangan yang masuk pada VIN maka VOUT akan mengeluarkan tegangan ~0 volt. Dan sebaliknya bila tegangan yang masuk pada VREF lebih kecil dari pada VIN maka VOUT akam mengeluarkan tegangan VSAT.

ControlSuhu:

mov a,dataSetting ; contoh dataSetting=50

mov b,dataADC ; contoh dataADC=30

clr c

subb a,b

jnz OnHeater

ret

OnHeater:

jc OffHeater

call HeaterOn ;Instruksi hidupkan heater

ret

OffHeater:

Call HeaterOff ;Instruksi matikan heater

ret

;

Pada instruksi tersebut diambil selisih antara dataSetting dan dataADC dengan menggunakan instruksi SUBB, pengurangan ini akan menghasilkan tiga keadaan yaitu: NOL, NEGATIF atau POSITIF. Hasil-hasil inilah yang harus dideteksi, keadaan NEGATIF dapat dideteksi dengan memantau bit C (carry ), keadaan NOL dapat dideteksi dengan memantau register A (accumulator).

Apabila diberikan keadaan input sesuai dengan contoh tersebut maka:

A=dataSetting=50

B=dataADC=30

SUBB A,B

A=50-30 =20 (keadaan POSITIF)

Sesuai dengan instruksi diatas maka program akan menuju ke Ret OnHeater, pada baris ini dilakukan proses pengujian keadaan, dengan instruksi JC, karena keadaan POSITIF maka C=0 (clear) sehingga program akan memanggil HeaterOn
Apabila diberikan keadaan input sesuai dengan contoh tersebut maka:

A=dataSetting=50

B=dataADC=50

SUBB A,B

A=50-50 =00 (keadaan NOL)
Sesuai dengan instruksi diatas maka program akan menuju ke Ret.

Apabila diberikan keadaan input sesuai dengan contoh tersebut maka:

A=dataSetting=50

B=dataADC=51

SUBB A,B

A=50-51 =-1 (keadaan NEGATIF)

Sesuai dengan instruksi diatas maka program akan menuju ke OnHeater, pada baris ini dilakukan proses pengujian keadaan, dengan instruksi JC, karena keadaan NEGATIF maka C=1 (clear) sehingga program akan memanggil label OffHeater
Pada percobaan 5.3. ini indikator heater On dan Off, ditunjukkan pada layar LCD Karakter pada baris 1. seperti yang ditunjukkan pada pemrograman berikut ini:

1. Pada saat langkah pemrograman posisikan saklar togle ke posisi PROG

2. Posisikan saklar togle ke RUN untuk mengaktifkan ADC0804 CS=0

3. Hubungkan modul Microcontroller Trainer dengan power supply +5V

4. Hubungkan modul Microcontroller Trainer dengan rangkaian programmer

5. Buka Program M-IDE Studio for MCS-51, sebagai editor dan compiler program

Gambar 5.11. Diagram alir (a) rutin utama (b) subrutin kontrol suhu pada percobaan 5.3.

dataSetting equ 30h

dataADC equ 31h

ratusan equ 32h

puluhan equ 33h

satuan equ 34h

org 0h

mov dataSetting,#50d; contoh datasetting=50

call init_lcd
start: 
call ADC

call ControlSuhu

call bin2dec

call Display2LCD

sjmp start

;

ControlSuhu:

mov a,dataSetting ; contoh dataSetting=50

mov b,dataADC ; contoh dataADC=30

clr c

subb a,b

jnz OnHeater

ret

OnHeater:

jc OffHeater

call HeaterOn ;Instruksi hidupkan heater

ret

OffHeater:

Call HeaterOff ;Instruksi matikan heater

ret

;

HeaterOn:

mov R1,#80h

call write_inst

mov R1,#'O'

call write_data

;

mov R1,#81h

call write_inst

mov R1,#'n'

call write_data

;

mov R1,#82h

call write_inst

mov R1,#' '

call write_data

ret

HeaterOff:

mov R1,#80h

call write_inst

mov R1,#'O'

call write_data

;

mov R1,#81h

call write_inst

mov R1,#'f'

call write_data

;

mov R1,#82h

call write_inst

mov R1,#'f'

call write_data

ret

;=======================================

;Subrutin ini untuk merubah data biner ke desimal

;menjadi 3 digit = ratusan-puluhan-satuan

;=======================================

Bin2Dec:

mov A,dataADC

mov b,#100d

div ab

mov ratusan,a

mov a,b

mov b,#10d

div ab

mov puluhan,a

mov satuan,b

ret

;=================================================

;Subrutin untuk menampilkan data ke LCD character 2 x16

;pada DDRAM 0C9 0CA 0CB untukratusan, puluhan, and satuan

;=================================================

Display2LCD:

mov r1,#0c0h

call write_inst

mov a,ratusan

add a,#30h

mov r1,a

call write_data

;

mov r1,#0c1h

call write_inst

mov a,puluhan

add a,#30h

mov r1,a

call write_data

;

mov r1,#0c2h

call write_inst

mov a,satuan

add a,#30h

mov r1,a

call write_data

ret

;==============================================

;Subrutin ini digunakan untuk mengambil data ADC MUX X0

;==============================================

ADC: clr P2.6

clr P2.7

clr P3.3

nop

nop

nop

setb P3.3

eoc: 
jb P3.2,eoc

clr P3.4

mov A,P1

mov dataADC,A

setb P3.4

ret

;

Init_lcd:

mov r1,#00000001b ;Display clear

call write_inst

mov r1,#00111000b ;Function set,Data 8 bit,2 line font 5x7

call write_inst

mov r1,#00001100b ;Display on, cursor off,cursor blink off

call write_inst

mov r1,#00000110b ;Entry mode, Set increment

call write_inst

ret

;

write_inst:

clr P3.6 ; RS = P2.0 = 0, write mode instruction

mov P0,R1 ; D7 s/d D0 = P0 = R1

setb P3.7 ; EN = 1 = P2.1

call delay ; call delay time

clr P3.7 ; EN = 0 = P2.1

ret

;

Write_data:

setb P3.6 ; RS = P2.0 = 1, write mode data

mov P0,R1 ; D7 s/d D0 = P0 = R1

setb P3.7 ; EN = 1 = P2.1

call delay ; call delay time

clr p3.7 ; EN = 0 = P2.1

ret

;

delay: 
mov R0,#0

delay1:mov R2,#0fh

djnz R2,$

djnz R0,delay1

ret

end

6. Simpanlah program yang anda ketik dan beri nama : prog53a.asm

7. Pada program MIDE tersebut pilih Build /F9 untuk melakukan kompilasi program dari *.asm ke *.hex.

8. Lakukan pemrograman mikrokontroller dengan menggunakan Program ISP Software (Lihat Petunjuk Penggunaan)

9. Lakukan modifikasi pada program tersebut dengan manambahkan kata SUHU: , pada Baris 2 diikuti dengan data ADC.

Percobaan 5.4. Kalibrasi dataADC ke suhu dengan menggunakan metode Look Up Table.

Kenapa kita membutuhkan look up table?: Look up table adalah suatu cara yang digunakan untuk menghindari proses perkalian dan pembagian yang bertele-tele dan memusingkan bila dilakukan dengan menggunakan bahasa assembly, yang tentunya harus dilakukan bila kita akan kalibrasi suatu alat ukur. Contoh kalibrasi Termometer dengan menggunakan persamaan persamaan berikut ini: Suhu = DataADC * 100/ 255 oC.
Contoh table untuk konversi data ke besaran suhu ( dengan menggunakan program Microsoft Excell ). Karena data decimal maksimal adalah 255 dan suhu maksimal 100 maka Data look up tablenya adalah 255/100.

0 0 0 0 0 0
1 0.4 0 0 0 4
2 0.8 0 0 0 8
3 12 0 0 1 2
4 16 0 0 1 6
5 20 0 0 2 0
6 24 0 0 2 4
7 27 0 0 2 7
8 3.1 0 0 3 1
9 35 0 0 3 5

248 97.3 0 9 7 3
249 97.6 0 9 7 6
250 98.0 0 9 8 0
251 98.4 0 9 8 4
252 98.8 0 9 8 8
253 99.2 0 9 9 2
254 99.6 0 9 9 6
255 100.0 1 0 0 0





Pada percobaan 5.4, kalibrasi dilakukan untuk perubahan range desimal (0 s/d 255) menjadi range suhu(000.0 s/d 100.0 oC)

1. Pada saat langkah pemrograman posisikan saklar togle ke posisi PROG

2. Posisikan saklar togle ke RUN untuk mengaktifkan ADC0804 CS=0

3. Hubungkan modul Microcontroller Trainer dengan power supply +5V

4. Hubungkan modul Microcontroller Trainer dengan rangkaian programmer

5. Buka Program M-IDE Studio for MCS-51, sebagai editor dan compiler program

dataADC equ 30h

org 0h

start: 
call ADC

call Display2SevenSegmen

sjmp start

;

;=======================================

;Subrutin ini digunakan untuk mengambil data ADC

;========================================

ADC: 
clr P2.6

clr P2.7

clr P3.3

nop

nop

nop

setb P3.3

eoc: 
jb P3.2,eoc

clr P3.4

mov A,P1

mov dataADC,A

setb P3.4

ret

;

Display2SevenSegmen:

mov DPTR,#ratusan ; DPTR = [ Ratusan ]

mov A,DataADC ; A = [DataADC]

movc A,@A+DPTR ; A = [A+DPTR]

mov DPTR,#Data7segmen ; DPTR = [Data7Segmen]

movc A,@A+DPTR ; A = [A+DPTR]

mov P0,A ; Copy A ke P0

Clr P3.5 ; Decoder, A=1,

Clr P3.6 ; B=0

Setb P3.7 ; dan C=1

call delay ; Panggil waktu tunda

;

mov DPTR,#puluhan ; DPTR = [ Puluhan ]

mov A,DataADC ; A = DataADC

movc A,@A+DPTR ; A =[ A+DPTR]

mov DPTR,#Data7segmen ; DPTR = [Data7Segmen]

movc A,@A+DPTR ; A = [A+DPTR]

mov P0,A

Setb P3.5

Clr P3.6

Setb P3.7

call delay

;

mov DPTR,#Satuan

mov A,DataADC

movc A,@A+DPTR

mov DPTR,#Data7segmen

movc A,@A+DPTR

mov P0,A

Clr P3.5 ;

Setb P3.6

Setb P3.7

call delay

;

mov DPTR,#Pecahan

mov A,DataADC

movc A,@A+DPTR

mov DPTR,#Data7segmen

movc A,@A+DPTR

mov P0,A

Setb P3.5 ;

Setb P3.6

Setb P3.7

call delay

ret

;

delay: 
mov R0,#0

delay1:mov R2,#0fh

djnz R2,$

djnz R0,delay1

ret

;

Pecahan:

db 0,4,8,2,6,0,4,7,1,5,9,3,7,1,5,9,3,7,1,5,8,2,6,0,4,8,2,6,0,4,8,2,5,9,3,7,1,5

db 9,3,7,1,5,9,3,6,0,4,8,2,6,0,4,8,2,6,0,4,7,1,5,9,3,7,1,5,9,3,7,1,5,8,2,6,0,4

db 8,2,6,0,4,8,2,5,9,3,7,1,5,9,3,7,1,5,9,3,6,0,4,8,2,6,0,4,8,2,6,0,4,7,1,5,9,3

db 7,1,5,9,3,7,1,5,8,2,6,0,4,8,2,6,0,4,8,2,5,9,3,7,1,5,9,3,7,1,5,9,3,6,0,4,8,2

db 6,0,4,8,2,6,0,4,7,1,5,9,3,7,1,5,9,3,7,1,5,8,2,6,0,4,8,2,6,0,4,8,2,5,9,3,7,1

db 5,9,3,7,1,5,9,3,6,0,4,8,2,6,0,4,8,2,6,0,4,7,1,5,9,3,7,1,5,9,3,7,1,5,8,2,6,0

db 4,8,2,6,0,4,8,2,5,9,3,7,1,5,9,3,7,1,5,9,3,6,0,4,8,2,6,0

;

Satuan:

db 0,0,0,1,1,2,2,2,3,3,3,4,4,5,5,5,6,6,7,7,7,8,8,9,9,9,0,0,1,1,1,2,2,2,3,3,4,4

db 4,5,5,6,6,6,7,7,8,8,8,9,9,0,0,0,1,1,2,2,2,3,3,3,4,4,5,5,5,6,6,7,7,7,8,8,9,9

db 9,0,0,1,1,1,2,2,2,3,3,4,4,4,5,5,6,6,6,7,7,8,8,8,9,9,0,0,0,1,1,2,2,2,3,3,3,4

db 4,5,5,5,6,6,7,7,7,8,8,9,9,9,0,0,1,1,1,2,2,2,3,3,4,4,4,5,5,6,6,6,7,7,8,8,8,9

db 9,0,0,0,1,1,2,2,2,3,3,3,4,4,5,5,5,6,6,7,7,7,8,8,9,9,9,0,0,1,1,1,2,2,2,3,3,4

db 4,4,5,5,6,6,6,7,7,8,8,8,9,9,0,0,0,1,1,2,2,2,3,3,3,4,4,5,5,5,6,6,7,7,7,8,8,9

db 9,0,0,0,1,1,2,2,2,3,3,3,4,4,5,5,5,6,6,7,7,7,8,8,9,9,9,0

;

puluhan:

db 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,1,1,1,1,1,1,1,1,1,1,1,1

db 1,1,1,1,1,1,1,1,1,1,1,1,1,2,2,2,2,2,2,2,2,2,2,2,2,2,2,2,2,2,2,2,2,2,2,2,2,2

db 2,3,3,3,3,3,3,3,3,3,3,3,3,3,3,3,3,3,3,3,3,3,3,3,3,3,4,4,4,4,4,4,4,4,4,4,4,4

db 4,4,4,4,4,4,4,4,4,4,4,4,4,4,5,5,5,5,5,5,5,5,5,5,5,5,5,5,5,5,5,5,5,5,5,5,5,5

db 5,6,6,6,6,6,6,6,6,6,6,6,6,6,6,6,6,6,6,6,6,6,6,6,6,6,7,7,7,7,7,7,7,7,7,7,7,7

db 7,7,7,7,7,7,7,7,7,7,7,7,7,7,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8

db 8,8,9,9,9,9,9,9,9,9,9,9,9,9,9,9,9,9,9,9,9,9,9,9,9,9,9,0

;

Ratusan:

db 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0

db 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0

db 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0

db 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0

db 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0

db 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0

db 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,1

;

Data7segmen:

db 11000000b,11111001b,10100100b,10110000b,10011001b

db 10010010b,10000010b,11111000b,10000000b,10010000b

;

end

Soal Kuis:

1. Pelajari mengenai pin konfigurasi dan fungsi dari ADC0804

2. Pelajari mengenai tegangan referensi, output data digital (bila diberi input tertentu)

3. Pelajari tentang instruksi assembly untuk keperluan ambil data ADC0804

4. Pelajari tentang interface ADC ke mikrokontroller




Subrutin waktu


Tunda/delay

